[bookmark: _GoBack]log using McGuireElementStataOutputTables2and3
*** VARIABLES
*** ccode: numerical code for country
*** codealp: alphabetical code for country
*** cname: name of country
*** year: year
*** rgdpnapc: GDP per capita in constant 2011 US$ using 2011 international price comparisons
*** wdi_mortinf: infant mortality according to WB WDI (downloaded 2 Nov 2017) via Quality of Government dataset (Teorell et al. 2018)
*** vdem_libdem: V-Dem liberal Democracy Index (downloaded 2 Nov 2017) Quality of Government dataset (Teorell et al. 2018)

*** Universe of cases: All countries in the world with data from 1970 to 2015 on GDP per capita at PPP in 2011 US$, the V-Dem liberal democracy index, and infant mortality. No missing data imputed.
*** Source: Jan Teorell et al., Quality of Government Standard Dataset, version Jan18. University of Gothenburg: The Quality of Government Institute, http://www.qog.pol.gu.se doi:10.18157/QoGStdJan18
*** Original source for GDP per capita at PPP in 2011 US$: Maddison Project Database 2018, Stata version, accessed 24 January 2018 at https://www.rug.nl/ggdc/historicaldevelopment/maddison/data/mpd2018.dta.
*** Original source for infant mortality: World Bank World Development Indicators accessed 2 November 2018.
*** Original source for V-Dem liberal Democracy Index: Varieties of Democracy (V-Dem) Project country-year/country-date dataset v7.1, downloaded 17 October 2017.

*** stipulate the panel and time variables
tsset ccode year

*** generate dummy variables for country and year
tab ccode, gen(ctry)
tab year, gen(yr)

*** generate a time trend variable
gen time = year-1970

*** take the natural logarithms of GDP per capita and infant mortality
gen lrgdpnapc = ln(rgdpnapc)
gen lwdi_mortinf = ln(wdi_mortinf)

*** generate lagged variables for use with Stata routines that cannot handle factor variables
gen lag1wdi_mortinf = l1.wdi_mortinf
gen lag1lwdi_mortinf = l1.wdi_mortinf
gen lag1vdem_libdem = l1.vdem_libdem
gen lag2vdem_libdem = l2.vdem_libdem
gen lag3vdem_libdem = l3.vdem_libdem
gen lag4vdem_libdem = l4.vdem_libdem
gen lag5vdem_libdem = l5.vdem_libdem

*** Summarize all variables
summarize ctry* yr*

***Diagnostics for reduced data set (1970-2015)

*** 1. Wald test for joint significance of province and year dummies to ascertain the advisability of including these dummies in the regressions

*** 2. Hausman tests for consistency of random effects model (if P < .05, random effects is inconsistent)

xtreg wdi_mortinf lrgdpnapc l1.vdem_libdem, fe
estimates store fixed
xtreg wdi_mortinf lrgdpnapc l1.vdem_libdem, re 
estimates store random
hausman fixed random, sigmamore
estimates store clear

xtreg lwdi_mortinf lrgdpnapc l1.vdem_libdem, fe
estimates store fixed
xtreg lwdi_mortinf lrgdpnapc l1.vdem_libdem, re 
estimates store random
hausman fixed random, sigmamore
estimates store clear

*** 3. Likelihood ratio test to see if intercepts vary across groups (if P < .05, intercepts vary and fixed effects is the appropriate model)

xtreg wdi_mortinf lrgdpnapc l1.vdem_libdem, i(ccode) mle
xtreg lwdi_mortinf lrgdpnapc l1.vdem_libdem, i(ccode) mle

*** 4. Modified Wald statistic to detect the presence of groupwise heteroskedasticity in the residuals of a fixed effect regression model (if P < .05, groupwise heteroskedasticity is present).

xtreg wdi_mortinf lrgdpnapc l1.vdem_libdem yr*, fe vce(r)
xttest3
xtreg lwdi_mortinf lrgdpnapc l1.vdem_libdem yr*, fe vce(r)
xttest3


*** 5. Wooldridge test for autocorrelation in panel data (if P < .05, error term is characterized by first-order autocorrelation)

xtserial wdi_mortinf lrgdpnapc lag1vdem_libdem
xtserial lwdi_mortinf lrgdpnapc lag1vdem_libdem

*** 6. Pesaran test for cross-sectional correlation in fixed effects model (if P < .05, error term is characterized by cross-sectional correlation)

xtreg wdi_mortinf lrgdpnapc l1.vdem_libdem, fe
xtcsd, pesaran
xtreg lwdi_mortinf lrgdpnapc l1.vdem_libdem, fe
xtcsd, pesaran


***Table 2 Regressions

*** Models 1 and 2: GLS with time (year) as well as unit (country) fixed effects and country-clustered standard errors 

xtreg wdi_mortinf lrgdpnapc L.vdem_libdem yr*, fe cluster(ccode)
xtreg lwdi_mortinf lrgdpnapc L.vdem_libdem yr*, fe cluster(ccode)

*** Models 3 and 4: pooled OLS with country and year fixed effects and panel-corrected standard errors

xtpcse wdi_mortinf lrgdpnapc L.vdem_libdem yr* ctry*, c(ar1)
xtpcse lwdi_mortinf lrgdpnapc L.vdem_libdem yr* ctry*, c(ar1)

*** Models 5 and 6: pooled OLS with country and year fixed effects, a lagged dependent variable, and panel-corrected standard errors

xtpcse wdi_mortinf lrgdpnapc L.vdem_libdem L.wdi_mortinf yr* ctry*, c(ar1)
xtpcse lwdi_mortinf lrgdpnapc L.vdem_libdem L.lwdi_mortinf yr* ctry*, c(ar1)


***Table 3 Regressions

*** Models 7 and 8: GLS with unit (country) but not time fixed effects and country-clustered standard errors 

xtreg wdi_mortinf lrgdpnapc L.vdem_libdem, fe cluster(ccode)
xtreg lwdi_mortinf lrgdpnapc L.vdem_libdem, fe cluster(ccode)

*** Models 9 and 10: pooled OLS with country and year fixed effects and Driscoll-Kraay standard errors

xtscc wdi_mortinf lrgdpnapc lag1vdem_libdem yr*, fe
xtscc lwdi_mortinf lrgdpnapc lag1vdem_libdem yr*, fe

*** Models 11 and 12: pooled OLS with country and year fixed effects, a lagged dependent variable, and Driscoll-Kraay standard errors

xtscc wdi_mortinf lrgdpnapc lag1vdem_libdem lag1wdi_mortinf yr*, fe
xtscc lwdi_mortinf lrgdpnapc lag1vdem_libdem lag1wdi_mortinf yr*, fe
